NAG Library Routine Document

G05HKF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G05HKF generates a given number of terms of a type I AGARCH(p, q) process (see Engle and Ng (1993)).

2 Specification

```
 SUBROUTINE G05HKF(DIST, NUM, IP, IQ, THETA, GAMMA, DF, HT, ET, FCALL,
1                  RVEC, IGEN, ISEED, RWSAV, IFAIL)
 INTEGER           NUM, IP, IQ, IGEN, ISEED(4), IFAIL
 double precision  THETA(IQ+IP+1), GAMMA, DF, HT(NUM), ET(NUM), RVEC(40),
1                  RWSAV(9)
 LOGICAL           FCALL
 CHARACTER*1       DIST
```

3 Description

A type I AGARCH(p, q) process can be represented by:

$$h_t = \alpha_0 + \sum_{i=1}^{q} \alpha_i (\epsilon_{t-i} + \gamma)^2 + \sum_{i=1}^{p} \beta_i h_{t-i}, \qquad t = 1, 2, \ldots, T$$

where $\epsilon_t \mid \psi_{t-1} = N(0, h_t)$ or $\epsilon_t \mid \psi_{t-1} = S_t(df, h_t)$. Here S_t is a standardized Student's t-distribution with df degrees of freedom and variance h_t, T is the number of observations in the sequence, ϵ_t is the *observed* value of the GARCH(p, q) process at time t, h_t is the conditional variance at time t, and ψ_t the set of all information up to time t. Symmetric GARCH sequences are generated when γ is zero, otherwise asymmetric GARCH sequences are generated with γ specifying the amount by which positive (or negative) shocks are to be enhanced.

One of the initialization routines G05KBF (for a repeatable sequence if computed sequentially) or G05KCF (for a non-repeatable sequence) must be called prior to the first call to G05HKF.

4 References

Bollerslev T (1986) Generalised autoregressive conditional heteroskedasticity *Journal of Econometrics* **31** 307–327

Engle R (1982) Autoregressive conditional heteroskedasticity with estimates of the variance of United Kingdom inflation *Econometrica* **50** 987–1008

Engle R and Ng V (1993) Measuring and Testing the Impact of News on Volatility *Journal of Finance* **48** 1749–1777

Hamilton J (1994) *Time Series Analysis* Princeton University Press

5 Parameters

1: DIST – CHARACTER*1 *Input*

On entry: the type of distribution to use for ϵ_t.

DIST = 'N'
A Normal distribution is used.

DIST = 'T'
A Student's t-distribution is used.

Constraint: DIST = 'N' or 'T'.

2: NUM – INTEGER *Input*

On entry: T, the number of terms in the sequence.

Constraint: NUM > 0.

3: IP – INTEGER *Input*

On entry: the number of coefficients, β_i, for $i = 1, 2, \ldots, p$.

Constraints:

IQ + IP + 1 $\leq$ 20;
IP $\geq$ 0.

4: IQ – INTEGER *Input*

On entry: the number of coefficients, α_i, for $i = 1, 2, \ldots, q$.

Constraints:

IQ + IP + 1 $\leq$ 20;
IQ $\geq$ 1.

5: THETA(IQ + IP + 1) – ***double precision*** array *Input*

On entry: the first element must contain the coefficient α_o, the next IQ elements must contain the coefficients α_i, for $i = 1, 2, \ldots, q$. The remaining IP elements must contain the coefficients β_j, for $j = 1, 2, \ldots, p$.

6: GAMMA – ***double precision*** *Input*

On entry: the asymmetry parameter γ for the GARCH(p, q) sequence.

7: DF – ***double precision*** *Input*

On entry: the number of degrees of freedom for the Student's t-distribution.

If DIST = 'N', DF is not referenced.

Constraint: if DIST = 'T', DF > 2.

8: HT(NUM) – ***double precision*** array *Output*

On exit: the conditional variances h_t, for $t = 1, 2, \ldots, T$, for the GARCH(p, q) sequence.

9: ET(NUM) – ***double precision*** array *Output*

On exit: the observations ϵ_t, for $t = 1, 2, \ldots, T$, for the GARCH(p, q) sequence.

10: FCALL – LOGICAL *Input*

On entry: if FCALL = .TRUE., a new sequence is to be generated, otherwise a given sequence is to be continued using the information in RVEC.

11: RVEC(40) – ***double precision*** array *Input/Output*

On entry: the array contains information required to continue a sequence if FCALL = .FALSE..

On exit: contains information that can be used in a subsequent call of G05HKF, with FCALL = .FALSE..

12: IGEN – INTEGER *Input*

On entry: must contain the identification number for the generator to be used to return a pseudorandom number and should remain unchanged following initialization by a prior call to G05KBF or G05KCF.

13: ISEED(4) – INTEGER array *Communication Array*

On entry: contains values which define the current state of the selected generator.

On exit: contains updated values defining the new state of the selected generator.

14: RWSAV(9) – ***double precision*** array *Communication Array*

15: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, −1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value −1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value −1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or −1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, IP < 0,
or IQ < 1,
or DF < 2,
or NUM ≤ 0,
or DIST $\neq$ 'N', and DIST $\neq$ 'T',
or IQ + IP + 1 > 20.

7 Accuracy

Not applicable.

8 Further Comments

None.

9 Example

This example uses G05HKF to generate 1500 data points, with known process parameters θ for the following two time series:

(i) an ARCH(3) sequence with Normally distributed residuals;

(ii) a GARCH(1, 2) sequence with Student's t-distributed residuals.

Here G05HKF is initially called, with the output discarded, to eliminate 'start-up effects' in these sequences. The process parameter estimates, $\hat{\theta}$, are then obtained using G13FAF, and compared with their true values, θ. Finally a four step ahead volatility estimate is computed using G13FBF.

9.1 Program Text

```
*     G05HKF Example Program Text
*     Mark 22 Release. NAG Copyright 2008.
*     .. Parameters ..
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
      INTEGER          NPARMX, LDCOVR, NUM
      DOUBLE PRECISION ZERO
      PARAMETER        (NPARMX=10,LDCOVR=NPARMX,NUM=1500,ZERO=0.0D0)
      INTEGER          NUM1, NREGMX
      PARAMETER        (NUM1=3000,NREGMX=10)
*     .. Local Scalars ..
      DOUBLE PRECISION DF, FAC1, GAMMA, HP, LGF, MEAN, TOL, XTERM
      INTEGER          I, IFAIL, IGEN, IP, IQ, ISYM, K, LDX, LWORK,
     +                 MAXIT, MN, NPAR, NPAR2, NREG, NT
      LOGICAL          FCALL
      CHARACTER        DIST
*     .. Local Arrays ..
      DOUBLE PRECISION BX(10), COVR(LDCOVR,NPARMX), CVAR(100),
     +                 ETM(NUM1), HT(NUM1+10), HTM(NUM1), PARAM(NPARMX),
     +                 RVEC(40), RWSAV(9), SC(NPARMX), SE(NPARMX),
     +                 THETA(NPARMX), WORK(NUM1*3+NPARMX+NREGMX*NUM1+20*
     +                 20+1), X(NUM1,10), YT(NUM1+10)
      INTEGER          ISEED(4)
      LOGICAL          COPTS(2)
*     .. External Subroutines ..
      EXTERNAL         G05HKF, G05KBF, G13FAF, G13FBF
*     .. Intrinsic Functions ..
      INTRINSIC        DBLE, SIN
*     .. Executable Statements ..
      WRITE (NOUT,*) 'G05HKF Example Program Results'
      WRITE (NOUT,*)
*
*     Initialize the generator
      ISEED(1) = 111
      IGEN = 0
      CALL G05KBF(IGEN,ISEED)
*
*     Example using Normal errors
      NREG = 0
      LDX = NUM1
      BX(1) = 1.5D0
      BX(2) = 2.5D0
      BX(3) = 3.0D0
      MEAN = 3.0D0
      DO 20 I = 1, NUM
         FAC1 = DBLE(I)*0.01D0
         X(I,1) = 0.01D0 + 0.7D0*SIN(FAC1)
         X(I,2) = 0.5D0 + FAC1*0.1D0
         X(I,3) = 1.0D0
   20 CONTINUE
*
      ISYM = 1
      MN = 1
      GAMMA = -0.4D0
      IP = 0
      IQ = 3
      PARAM(1) = 0.8D0
      PARAM(2) = 0.6D0
      PARAM(3) = 0.2D0
      PARAM(4) = 0.1D0
      NPAR = 1 + IQ + IP
      LWORK = NREG*NUM + 3*NUM + NPAR + ISYM + MN + NREG + 403
      DIST = 'N'
*
*     First call to routine
      FCALL = .TRUE.
      IFAIL = 0
      CALL G05HKF(DIST,NUM,IP,IQ,PARAM,GAMMA,DF,HT,YT,FCALL,RVEC,IGEN,
     +            ISEED,RWSAV,IFAIL)
```

```
*     Discard the first NUM values and call the routine again
      FCALL = .FALSE.
      IFAIL = 0
      CALL G05HKF(DIST,NUM,IP,IQ,PARAM,GAMMA,DF,HT,YT,FCALL,RVEC,IGEN,
     +            ISEED,RWSAV,IFAIL)
*
      DO 60 I = 1, NUM
         XTERM = ZERO
         DO 40 K = 1, NREG
            XTERM = XTERM + X(I,K)*BX(K)
   40    CONTINUE
         IF (MN.EQ.1) THEN
            YT(I) = MEAN + XTERM + YT(I)
         ELSE
            YT(I) = XTERM + YT(I)
         END IF
   60 CONTINUE
*
      DO 80 I = 1, NPAR
         THETA(I) = PARAM(I)*0.5D0
   80 CONTINUE
*
      IF (ISYM.EQ.1) THEN
         THETA(NPAR+ISYM) = GAMMA*0.5D0
      END IF
*
*     Fit a GARCH model to the generated data
      COPTS(1) = .TRUE.
      COPTS(2) = .TRUE.
      MAXIT = 200
      TOL = 1.0D-5
      NPAR2 = 1 + IQ + IP + ISYM + MN + NREG
      IFAIL = 0
      CALL G13FAF(DIST,YT,X,LDX,NUM,IP,IQ,NREG,MN,ISYM,NPAR2,THETA,SE,
     +            SC,COVR,LDCOVR,HP,ETM,HTM,LGF,COPTS,MAXIT,TOL,WORK,
     +            LWORK,IFAIL)
*     Display the output
      WRITE (NOUT,*)
      WRITE (NOUT,*) 'Normal distribution'
      WRITE (NOUT,*)
      WRITE (NOUT,*) '             Correct    Parameter   Standard'
      WRITE (NOUT,*) '             values     estimates    errors'
      DO 100 I = 1, NPAR
         WRITE (NOUT,99998) 'PARAM(', I, ')', PARAM(I), THETA(I), SE(I)
  100 CONTINUE
      IF (ISYM.EQ.1) THEN
         WRITE (NOUT,99999) 'Gamma   ', GAMMA, THETA(NPAR+1), SE(NPAR+1)
      END IF
      IF (MN.EQ.1) THEN
         WRITE (NOUT,99999) 'Mean    ', MEAN, THETA(NPAR+ISYM+1),
     +     SE(NPAR+ISYM+1)
      END IF
      DO 120 I = 1, NREG
         WRITE (NOUT,99998) 'BX(', I, ')   ', BX(I),
     +     THETA(NPAR+ISYM+MN+I), SE(NPAR+ISYM+MN+I)
  120 CONTINUE
*
      NT = 4
      IFAIL = 0
      CALL G13FBF(NUM,NT,IP,IQ,THETA,GAMMA,CVAR,HTM,ETM,IFAIL)
      WRITE (NOUT,*)
      WRITE (NOUT,99997) 'Volatility forecast = ', CVAR(NT)
      WRITE (NOUT,*)
*
*     Example using Students T errors
      DIST = 'T'
      NREG = 2
      MN = 1
      DF = 4.1D0
      IP = 1
      IQ = 2
```

```
      ISYM = 1
      GAMMA = -0.2D0
*
      NPAR = IQ + IP + 1
      LWORK = NREG*NUM + 3*NUM + NPAR + ISYM + MN + NREG + 404
*
      PARAM(1) = 0.1D0
      PARAM(2) = 0.2D0
      PARAM(3) = 0.3D0
      PARAM(4) = 0.4D0
*
      IGEN = 0
      ISEED(1) = 111
      CALL G05KBF(IGEN,ISEED)
*
*     First call to the routine
      IFAIL = 0
      FCALL = .TRUE.
      CALL G05HKF(DIST,NUM,IP,IQ,PARAM,GAMMA,DF,HT,YT,FCALL,RVEC,IGEN,
     +            ISEED,RWSAV,IFAIL)
*     Second call to the routine
      FCALL = .FALSE.
      CALL G05HKF(DIST,NUM,IP,IQ,PARAM,GAMMA,DF,HT,YT,FCALL,RVEC,IGEN,
     +            ISEED,RWSAV,IFAIL)
*     Discard the previously generated values, and call the routine again
      FCALL = .FALSE.
      CALL G05HKF(DIST,NUM,IP,IQ,PARAM,GAMMA,DF,HT,YT,FCALL,RVEC,IGEN,
     +            ISEED,RWSAV,IFAIL)
*
      DO 160 I = 1, NUM
         XTERM = ZERO
         DO 140 K = 1, NREG
            XTERM = XTERM + X(I,K)*BX(K)
  140    CONTINUE
         IF (MN.EQ.1) THEN
            YT(I) = MEAN + XTERM + YT(I)
         ELSE
            YT(I) = XTERM + YT(I)
         END IF
  160 CONTINUE
*
      COPTS(1) = .TRUE.
      COPTS(2) = .TRUE.
*
      MAXIT = 200
      TOL = 1.0D-5
*
      DO 180 I = 1, NPAR
         THETA(I) = PARAM(I)*0.5D0
  180 CONTINUE
*
      THETA(NPAR+ISYM) = GAMMA*0.5D0
      THETA(NPAR+ISYM+1) = DF*0.65D0
*
      NPAR2 = 2 + IP + IQ + ISYM + MN + NREG
      IFAIL = 0
      CALL G13FAF(DIST,YT,X,LDX,NUM,IP,IQ,NREG,MN,ISYM,NPAR2,THETA,SE,
     +            SC,COVR,LDCOVR,HP,ETM,HTM,LGF,COPTS,MAXIT,TOL,WORK,
     +            LWORK,IFAIL)
*
      WRITE (NOUT,*)
      WRITE (NOUT,*) 'Students t distribution'
      WRITE (NOUT,*)
      WRITE (NOUT,*) '             Correct    Parameter   Standard'
      WRITE (NOUT,*) '             values     estimates    errors'
      DO 200 I = 1, NPAR
         WRITE (NOUT,99998) 'PARAM(', I, ')', PARAM(I), THETA(I), SE(I)
  200 CONTINUE
      IF (ISYM.EQ.1) THEN
         WRITE (NOUT,99999) 'Gamma   ', GAMMA, THETA(NPAR+ISYM),
     +     SE(NPAR+ISYM)
```

```
      END IF
      WRITE (NOUT,99999) 'DF      ', DF, THETA(NPAR+ISYM+1),
     +  SE(NPAR+ISYM+1)
      IF (MN.EQ.1) THEN
         WRITE (NOUT,99999) 'Mean    ', MEAN, THETA(NPAR+ISYM+2),
     +     SE(NPAR+ISYM+2)
      END IF
      DO 220 I = 1, NREG
         WRITE (NOUT,99998) 'BX(', I, ')   ', BX(I),
     +     THETA(NPAR+ISYM+1+MN+I), SE(NPAR+ISYM+1+MN+I)
  220 CONTINUE
*
      NT = 4
      IFAIL = 0
      CALL G13FBF(NUM,NT,IP,IQ,THETA,GAMMA,CVAR,HTM,ETM,IFAIL)
*
      WRITE (NOUT,*)
      WRITE (NOUT,99997) 'Volatility forecast = ', CVAR(NT)
*
99999 FORMAT (1X,A,3(F10.2,2X))
99998 FORMAT (1X,A,I1,A,3(F10.2,2X))
99997 FORMAT (1X,A,F12.2)
      END
```

9.2 Program Data

None.

9.3 Program Results

```
 G05HKF Example Program Results


 Normal distribution

              Correct    Parameter   Standard
              values     estimates    errors
 PARAM(1)      0.80        0.80        0.08
 PARAM(2)      0.60        0.62        0.06
 PARAM(3)      0.20        0.18        0.03
 PARAM(4)      0.10        0.09        0.02
 Gamma        -0.40       -0.51        0.07
 Mean          3.00        2.99        0.03

 Volatility forecast =         2.80


 Students t distribution

              Correct    Parameter   Standard
              values     estimates    errors
 PARAM(1)      0.10        0.09        0.02
 PARAM(2)      0.20        0.22        0.05
 PARAM(3)      0.30        0.28        0.09
 PARAM(4)      0.40        0.36        0.08
 Gamma        -0.20       -0.32        0.06
 DF            4.10        4.50        0.52
 Mean          3.00        3.02        0.05
 BX(1)         1.50        1.47        0.03
 BX(2)         2.50        2.46        0.03

 Volatility forecast =         0.42
```